

# NAG Library Routine Document

## E02AKF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

E02AKF evaluates a polynomial from its Chebyshev series representation, allowing an arbitrary index increment for accessing the array of coefficients.

### 2 Specification

```
SUBROUTINE E02AKF (NP1, XMIN, XMAX, A, IA1, LA, X, RESULT, IFAIL)
```

```
INTEGER NP1, IA1, LA, IFAIL
```

```
REAL (KIND=nag_wp) XMIN, XMAX, A(LA), X, RESULT
```

### 3 Description

If supplied with the coefficients  $a_i$ , for  $i = 0, 1, \dots, n$ , of a polynomial  $p(\bar{x})$  of degree  $n$ , where

$$p(\bar{x}) = \frac{1}{2}a_0 + a_1T_1(\bar{x}) + \dots + a_nT_n(\bar{x}),$$

E02AKF returns the value of  $p(\bar{x})$  at a user-specified value of the variable  $x$ . Here  $T_j(\bar{x})$  denotes the Chebyshev polynomial of the first kind of degree  $j$  with argument  $\bar{x}$ . It is assumed that the independent variable  $\bar{x}$  in the interval  $[-1, +1]$  was obtained from your original variable  $x$  in the interval  $[x_{\min}, x_{\max}]$  by the linear transformation

$$\bar{x} = \frac{2x - (x_{\max} + x_{\min})}{x_{\max} - x_{\min}}.$$

The coefficients  $a_i$  may be supplied in the array A, with any increment between the indices of array elements which contain successive coefficients. This enables the routine to be used in surface fitting and other applications, in which the array might have two or more dimensions.

The method employed is based on the three-term recurrence relation due to Clenshaw (see Clenshaw (1955)), with modifications due to Reinsch and Gentleman (see Gentleman (1969)). For further details of the algorithm and its use see Cox (1973) and Cox and Hayes (1973).

### 4 References

Clenshaw C W (1955) A note on the summation of Chebyshev series *Math. Tables Aids Comput.* **9** 118–120

Cox M G (1973) A data-fitting package for the non-specialist user *NPL Report NAC 40* National Physical Laboratory

Cox M G and Hayes J G (1973) Curve fitting: a guide and suite of algorithms for the non-specialist user *NPL Report NAC26* National Physical Laboratory

Gentleman W M (1969) An error analysis of Goertzel's (Watt's) method for computing Fourier coefficients *Comput. J.* **12** 160–165

## 5 Parameters

1: NP1 – INTEGER *Input*

*On entry:*  $n + 1$ , where  $n$  is the degree of the given polynomial  $p(\bar{x})$ .

*Constraint:*  $NP1 \geq 1$ .

2: XMIN – REAL (KIND=nag\_wp) *Input*

3: XMAX – REAL (KIND=nag\_wp) *Input*

*On entry:* the lower and upper end points respectively of the interval  $[x_{\min}, x_{\max}]$ . The Chebyshev series representation is in terms of the normalized variable  $\bar{x}$ , where

$$\bar{x} = \frac{2x - (x_{\max} + x_{\min})}{x_{\max} - x_{\min}}.$$

*Constraint:*  $XMIN < XMAX$ .

4: A(LA) – REAL (KIND=nag\_wp) array *Input*

*On entry:* the Chebyshev coefficients of the polynomial  $p(\bar{x})$ . Specifically, element  $i \times IA1 + 1$  must contain the coefficient  $a_i$ , for  $i = 0, 1, \dots, n$ . Only these  $n + 1$  elements will be accessed.

5: IA1 – INTEGER *Input*

*On entry:* the index increment of A. Most frequently, the Chebyshev coefficients are stored in adjacent elements of A, and IA1 must be set to 1. However, if, for example, they are stored in  $A(1), A(4), A(7), \dots$ , then the value of IA1 must be 3.

*Constraint:*  $IA1 \geq 1$ .

6: LA – INTEGER *Input*

*On entry:* the dimension of the array A as declared in the (sub)program from which E02AKF is called.

*Constraint:*  $LA \geq (NP1 - 1) \times IA1 + 1$ .

7: X – REAL (KIND=nag\_wp) *Input*

*On entry:* the argument  $x$  at which the polynomial is to be evaluated.

*Constraint:*  $XMIN \leq X \leq XMAX$ .

8: RESULT – REAL (KIND=nag\_wp) *Output*

*On exit:* the value of the polynomial  $p(\bar{x})$ .

9: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry,  $NP1 < 1$ ,  
 or  $IA1 < 1$ ,  
 or  $LA \leq (NP1 - 1) \times IA1$ ,  
 or  $XMIN \geq XMAX$ .

$IFAIL = 2$

X does not satisfy the restriction  $XMIN \leq X \leq XMAX$ .

## 7 Accuracy

The rounding errors are such that the computed value of the polynomial is exact for a slightly perturbed set of coefficients  $a_i + \delta a_i$ . The ratio of the sum of the absolute values of the  $\delta a_i$  to the sum of the absolute values of the  $a_i$  is less than a small multiple of  $(n + 1) \times \textit{machine precision}$ .

## 8 Further Comments

The time taken is approximately proportional to  $n + 1$ .

## 9 Example

Suppose a polynomial has been computed in Chebyshev series form to fit data over the interval  $[-0.5, 2.5]$ . The following program evaluates the polynomial at 4 equally spaced points over the interval. (For the purposes of this example, XMIN, XMAX and the Chebyshev coefficients are supplied in DATA statements. Normally a program would first read in or generate data and compute the fitted polynomial.)

### 9.1 Program Text

```

Program e02akfe

!      E02AKF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: e02akf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Real (Kind=nag_wp), Parameter      :: xmax = 2.5E0_nag_wp
      Real (Kind=nag_wp), Parameter      :: xmin = -0.5E0_nag_wp
      Integer, Parameter                  :: nout = 6, np1 = 7
      Integer, Parameter                  :: la = np1
      Real (Kind=nag_wp), Parameter      :: a(la) = (/2.53213E0_nag_wp,      &
        1.13032E0_nag_wp,0.27150E0_nag_wp,0.04434E0_nag_wp,0.00547E0_nag_wp, &
        0.00054E0_nag_wp,0.00004E0_nag_wp/)

!      .. Local Scalars ..
      Real (Kind=nag_wp)                  :: res, x
      Integer                              :: i, ifail, m

!      .. Intrinsic Procedures ..
      Intrinsic                            :: real

!      .. Executable Statements ..
      Write (nout,*) 'E02AKF Example Program Results'

      m = 4

```

```

Do i = 1, m
  x = (xmin*real(m-i,kind=nag_wp)+xmax*real(i-1,kind=nag_wp))/ &
    real(m-1,kind=nag_wp)

  ifail = 0
  Call e02akf(np1,xmin,xmax,a,1,la,x,res,ifail)

  If (i==1) Then
    Write (nout,*)
    Write (nout,*) ' I Argument Value of polynomial'
  End If

  Write (nout,99999) i, x, res
End Do

99999 Format (1X,I4,F10.4,4X,F9.4)
End Program e02akfe

```

## 9.2 Program Data

None.

## 9.3 Program Results

E02AKF Example Program Results

I	Argument	Value of polynomial
1	-0.5000	0.3679
2	0.5000	0.7165
3	1.5000	1.3956
4	2.5000	2.7183

